On Finding the Midpoint Locus of a Triangle in a Corner

(Or, as it is more affectionately known: Computing the Wetzel Pretzel)

Daniel Lichtblau
danl@wolfram.com
Wolfram Research, Inc.
100 Trade Center Dr.
Champaign IL 61820

ADG 2010 (Munich)
Technische Universität München
July 22−24, 2010

ABSTRACT

We are given an equilateral triangle with vertices constrained to lie in each of the three positive octant coordinate planes (colloquially, "a triangle in a corner"). We wish to describe the locus of points covered by the midpoint of the triangle, as the vertices range over configurations allowed by the above constraint. This locus comprises a solid region. We use numerical and graphical methods, and also computational algebra, to find the boundary surface and visualize this locus.

◄ | ►

The problem

We have an equilateral triangle. We position it so that each a floor and two walls, forming a corner, each have one vertex (we allow the pathological case where a vertex might be on an axis, that is, an edge where walls meet or where a wall meets the floor). We allow all motions that respect these constraints. For purposes of exposition and computation, we will take the corner to be the positive orthant in Cartesian space.

Question: How might we describe the locus of the triangle's midpoint?

This was posed in 2009 by Jack Wetzel and Richard Jerrard, emeriti from UIUC math Dept and, in particular, refugees from the Geometry Potpourri seminar that Jack organized and ran for around 30 years.)

Acknowledgement

This work was in earlier stages discussed by email with Jack and Wacharin Wichiramala. I also had assistance with some of the graphics from Michael Trott and Maxim Rytin. My thanks to all of them.

Why do we care about this problem?

- It is related to problems of historical interest e.g. "penny in a corner".
- It is related to problems in constraint geometry e.g. motion planning involving rigid bodies.
- It uses what I consider to be several interesting methods of computation.

Some observations

"Obvious" features:

- The midpoint locus is connected. If we give three points to the vertices, this (of course) determines the midpoint location. We can then reach any other allowed midpoint location by a continuous legal motion of the three vertices.
- It lies in the positive octant.

Less obvious:

- The locus is a solid.
- It has an enclosing surface that is comprised of parts of algebraic varieties.
- One will not find these varieties using only pencil and paper (which I regard as the primary justification for having this talk).

Setting up the math

1. Denote our midpoint as $M = \{x_m, y_m, z_m\}$.
2. Vertices will be $\{x_j, y_j, z_j\}$ for $1 \le j \le 3$.
3. Restrict vertices to coordinate planes by setting certain vertex coordinates to zero.
4. Quadratic equations for edge lengths enforce that the triangle be equilateral.
5. Linear equations relate midpoint coordinates to the vertex coordinates.

◂ | ▸

Equations

```
midpt = {xm, ym, zm};
coords = {x, y, z};
ptcoords = Map[Array[#, 3] &, coords];
pts = Transpose[ptcoords];
vals = MapIndexed[(#1[#2[[1]]] = 0) &,
   coords];
polys =
  Flatten[{Mean[pts] - midpt,
    Map[#.# - 1 &,
     Flatten[Table[pts[[j]] - pts[[k]],
       {j, Length[pts] - 1},
       {k, j + 1, Length[pts]}], 1]]}]
```

$$\left\{-x_m + \frac{1}{3}(x[2] + x[3]), -y_m + \frac{1}{3}(y[1] + y[3]), -z_m + \frac{1}{3}(z[1] + z[2]), -1 + x[2]^2 + y[1]^2 + (z[1] - z[2])^2, -1 + x[3]^2 + (y[1] - y[3])^2 + z[1]^2, -1 + (x[2] - x[3])^2 + y[3]^2 + z[2]^2\right\}$$

Note to self

You should take the time to open some of these cells, to show people what is going on.

Note to audience

Remind speaker of "Note to self".

◄ | ►

What to do with the equations?

We have six polynomials.

Interested in midpoint variables as functions of vertex coordinates. Regard latter as parameters.

We will eliminate three of them. Midpoint defined, implicitly, in terms of remaining three.

This explains why the locus is a solid: we have three degrees of freedom in the parameters that remain.

```
elims = {z[1], x[2], y[3]};
params =
  Complement[DeleteCases[Flatten[pts], 0],
   elims];
gbrat = GroebnerBasis[polys, midpt,
  elims, CoefficientDomain →
   RationalFunctions ,
  MonomialOrder → EliminationOrder]
```

$$\{1 - 18 z_m^2 - 6 x_m x[3] + 2 x[3]^2 + 6 y_m y[1] - 4 y[1]^2 + 18 z_m z[2] - 4 z[2]^2,$$
$$1 - 18 y_m^2 + 6 x_m x[3] - 4 x[3]^2 + 18 y_m y[1] - 4 y[1]^2 - 6 z_m z[2] + 2 z[2]^2,$$
$$1 - 18 x_m^2 + 18 x_m x[3] - 4 x[3]^2 - 6 y_m y[1] + 2 y[1]^2 + 6 z_m z[2] - 4 z[2]^2\}$$

How to visualize this locus

Idea: Take a lot of random vertex coordinates. Solve the system of equations. Discard all solutions that give coordinates either complex or out of range. The rest give a "point cloud" indicating the locus region.

Remark: During work on this project I messed up this part. Twice. It seems easy now, but it may have been the trickiest aspect to this work.

```
vertices = RandomReal[{0, 1}, {10 000, 3}];
substs = Map[Thread[params → #] &, vertices];
vars = Complement[Variables[polys], params];
```

Fnd all possible solutions.

```
Timing[
 solns =
   Flatten[Map[NSolve[polys /. #, vars] &,
     substs], 1];]
```

```
{372.271, Null}
```

Select those that meet configuration constraints.

```
mdptsols = Pick[midpt /. solns,
   Apply[And,
    Map[(Head[#] === Real && # ≥ 0) &,
     vars /. solns, {2}], {1}]];
```

◄ | ►

Visualize this locus...

Use symmetries of problem to get six times as many solutions (our vertices may be permuted by any element of S_3).



◄ | ►

Visualize this locus...

We can get a better picture by creating a region function. A point is deemed to be inside if it is sufficiently near to one of the points we computed. This gives a reasonable approximation to the correct region.

```
nf = Nearest[symmdptsols];
inRegion[pt : {_Real, _Real, _Real},
  eps_Real] :=
 TrueQ[Norm[nf[pt, 1][[1]] - pt] < eps]
```

```
reg = RegionPlot3D[inRegion[{x, y, z}, .02],
  {x, 0, .6}, {y, 0, .6}, {z, 0, .6},
  Mesh → False]
```



(Could improve picture by finding more points at the edges before creating the region function.)

◄ | ►

Solving algebraically

We will now see if we can deduce the boundary surface to this region. We first observe that at least some of it should arise as an extremal set. That is, for fixed x_m and y_m values of the midpoint, the surface is hit by extremal z_m values.

Regard our midpoint coordinates explicitly as functions of the (noneliminated) vertex parameters.

```
varsub = Map[# → #[Sequence @@ params] &,
   midpt];
fullrats = gbrat /. varsub;
reversesub = Map[Reverse, varsub];
ratderivs =
  Flatten[Map[D[fullrats, #] &, params]];
derivvars = Cases[Variables[ratderivs],
   Derivative[__][_][___]];
derivs =
  First[Solve[(ratderivs /. reversesub) == 0,
    derivvars]];
```

Quick view of what some of this has done:

```
ratderivs[[1]]
derivs[[1]]
```

$$4\,x[3] - 6\,x_m[x[3],\,y[1],\,z[2]] - 6\,x[3]\,(x_m)^{(1,0,0)}[x[3],\,y[1],\,z[2]] + 6\,y[1]\,(y_m)^{(1,0,0)}[x[3],\,y[1],\,z[2]] + 18\,z[2]\,(z_m)^{(1,0,0)}[x[3],\,y[1],\,z[2]] - 36\,z_m[x[3],\,y[1],\,z[2]]\,(z_m)^{(1,0,0)}[x[3],\,y[1],\,z[2]]$$

$$(x_m)^{(0,0,1)}[x[3],\,y[1],\,z[2]] \to \left(-9\,y_m\,z_m^2 + 3\,z_m^2\,y[1] + 12\,y_m\,z_m\,z[2] - 5\,z_m\,y[1]\,z[2] - 4\,y_m\,z[2]^2 + 2\,y[1]\,z[2]^2\right) \Big/ \left(3\,(-18\,x_m\,y_m\,z_m + 9\,y_m\,z_m\,x[3] + 9\,x_m\,z_m\,y[1] - 5\,z_m\,x[3]\,y[1] + 9\,x_m\,y_m\,z[2] - 5\,y_m\,x[3]\,z[2] - 5\,x_m\,y[1]\,z[2] + 3\,x[3]\,y[1]\,z[2])\right)$$

◄ | ►

Solving algebraically...

Now extremize z_m as an implicit function of the other two $\{x_m, \quad y_m\}$. We will set up as a Lagrange multiplier problem. We have a vector equation relating gradients of the above midpoint functions, using two multipliers.

Dimensional check: 2 new variables − 3 new equations removes one degree of freedom, so we go from region to surface.

```
grad[v_] :=
 Map[D[v /. varsub, #] &, params] /. derivs
auxpolys =
  Flatten[
   Together[grad[zm] - λ1 * grad[xm] -
     λ2 * grad[ym]]];
```

We now form, and show, the full set of polynomials. Except you cannot read it because the print is too small.

```
fullpolys = Join[gbrat, auxpolys]
```

$$\{1 - 18 z_m^2 - 6 x_m x[3] + 2 x[3]^2 + 6 y_m y[1] - 4 y[1]^2 + 18 z_m z[2] - 4 z[2]^2,$$

$$1 - 18 y_m^2 + 6 x_m x[3] - 4 x[3]^2 + 18 y_m y[1] - 4 y[1]^2 - 6 z_m z[2] + 2 z[2]^2,$$

$$1 - 18 x_m^2 + 18 x_m x[3] - 4 x[3]^2 - 6 y_m y[1] + 2 y[1]^2 + 6 z_m z[2] - 4 z[2]^2,$$

$$\Big(-9 x_m^2 y_m - 27 x_m y_m z_m \lambda_1 - 9 x_m^2 z_m \lambda_2 + 6 x_m y_m x[3] + 12 y_m z_m \lambda_1 x[3] + 12 x_m z_m \lambda_2 x[3] - y_m x[3]^2 - 4 z_m \lambda_2 x[3]^2 + 6 x_m^2 y[1] + 15 x_m z_m \lambda_1 y[1] - 5 x_m x[3] y[1] - 8 z_m \lambda_1 x[3] y[1] + x[3]^2 y[1] + 15 x_m y_m \lambda_1 z[2] + 3 x_m^2 \lambda_2 z[2] - 7 y_m \lambda_1 x[3] z[2] - 5 x_m \lambda_2 x[3] z[2] + 2 \lambda_2 x[3]^2 z[2] - 9 x_m \lambda_1 y[1] z[2] + 5 \lambda_1 x[3] y[1] z[2]\Big) \Big/ (3 (18 x_m y_m z_m - 9 y_m z_m x[3] - 9 x_m z_m y[1] + 5 z_m x[3] y[1] - 9 x_m y_m z[2] + 5 y_m x[3] z[2] + 5 x_m y[1] z[2] - 3 x[3] y[1] z[2])),$$

$$\Big(9 x_m y_m^2 + 9 y_m^2 z_m \lambda_1 - 27 x_m y_m z_m \lambda_2 - 3 y_m^2 x[3] + 15 y_m z_m \lambda_2 x[3] - 12 x_m y_m y[1] - 6 y_m z_m \lambda_1 y[1] + 12 x_m z_m \lambda_2 y[1] + 5 y_m x[3] y[1] - 7 z_m \lambda_2 x[3] y[1] + 4 x_m y[1]^2 + z_m \lambda_1 y[1]^2 - 2 x[3] y[1]^2 - 6 y_m^2 \lambda_1 z[2] + 15 x_m y_m \lambda_2 z[2] - 9 y_m \lambda_2 x[3] z[2] + 5 y_m \lambda_1 y[1] z[2] - 8 x_m \lambda_2 y[1] z[2] + 5 \lambda_2 x[3] y[1] z[2] - \lambda_1 y[1]^2 z[2]\Big) \Big/ (3 (18 x_m y_m z_m - 9 y_m z_m x[3] - 9 x_m z_m y[1] + 5 z_m x[3] y[1] - 9 x_m y_m z[2] + 5 y_m x[3] z[2] + 5 x_m y[1] z[2] - 3 x[3] y[1] z[2])),$$

$$\Big(27 x_m y_m z_m - 9 y_m z_m^2 \lambda_1 + 9 x_m z_m^2 \lambda_2 - 15 y_m z_m x[3] - 6 z_m^2 \lambda_2 x[3] - 15 x_m z_m y[1] + 3 z_m^2 \lambda_1 y[1] + 9 z_m x[3] y[1] - 12 x_m y_m z[2] + 12 y_m z_m \lambda_1 z[2] - 6 x_m z_m \lambda_2 z[2] + 8 y_m x[3] z[2] + 5 z_m \lambda_2 x[3] z[2] + 7 x_m y[1] z[2] - 5 z_m \lambda_1 y[1] z[2] - 5 x[3] y[1] z[2] - 4 y_m \lambda_1 z[2]^2 + x_m \lambda_2 z[2]^2 - \lambda_2 x[3] z[2]^2 + 2 \lambda_1 y[1] z[2]^2\Big) \Big/ (3 (18 x_m y_m z_m - 9 y_m z_m x[3] - 9 x_m z_m y[1] + 5 z_m x[3] y[1] - 9 x_m y_m z[2] + 5 y_m x[3] z[2] + 5 x_m y[1] z[2] - 3 x[3] y[1] z[2]))\}$$

◄ | ►

Solving algebraically...

Next step: find the polynomial that implicitly relates the midpoint coordinates. This means we eliminate the vertex parameters and Lagrange multipliers from the polynomials above. For this we use a Gröbner basis, with some settings especially adept at handling this type of problem. Without which, we might still be awaiting our surface equation...

```
Timing[
 implicit =
  First[GroebnerBasis[fullpolys, midpt,
    Join[{λ1, λ2}, params], Sort → True,
    Method → {"GroebnerWalk",
      "EarlyEliminate" → True}]]]
```

$$\{6.739,\ 16 - 672\,x_m^2 + 9432\,x_m^4 - 37\,800\,x_m^6 - 155\,439\,x_m^8 + 680\,400\,x_m^{10} + 3\,055\,968\,x_m^{12} + 3\,919\,104\,x_m^{14} + 1\,679\,616\,x_m^{16} - 672\,y_m^2 + 19\,728\,x_m^2\,y_m^2 - 245\,592\,x_m^4\,y_m^2 + 1\,821\,852\,x_m^6\,y_m^2 - 4\,412\,880\,x_m^8\,y_m^2 - 25\,649\,136\,x_m^{10}\,y_m^2 - 15\,396\,480\,x_m^{12}\,y_m^2 + 8\,398\,080\,x_m^{14}\,y_m^2 + 9432\,y_m^4 - 245\,592\,x_m^2\,y_m^4 + 1\,540\,134\,x_m^4\,y_m^4 - 9\,082\,368\,x_m^6\,y_m^4 + 119\,322\,720\,x_m^8\,y_m^4 - 41\,150\,592\,x_m^{10}\,y_m^4 + 16\,796\,160\,x_m^{12}\,y_m^4 - 37\,800\,y_m^6 + 1\,821\,852\,x_m^2\,y_m^6 - 9\,082\,368\,x_m^4\,y_m^6 - 112\,091\,040\,x_m^6\,y_m^6 - 19\,035\,648\,x_m^8\,y_m^6 + 18\,475\,776\,x_m^{10}\,y_m^6 - 155\,439\,y_m^8 - 4\,412\,880\,x_m^2\,y_m^8 + 119\,322\,720\,x_m^4\,y_m^8 - 19\,035\,648\,x_m^6\,y_m^8 + 16\,796\,160\,x_m^8\,y_m^8 + 680\,400\,y_m^{10} - 25\,649\,136\,x_m^2\,y_m^{10} - 41\,150\,592\,x_m^4\,y_m^{10} +$$

$18\,475\,776\, x_m^6\, y_m^{10} + 3\,055\,968\, y_m^{12} -$
$15\,396\,480\, x_m^2\, y_m^{12} + 16\,796\,160\, x_m^4\, y_m^{12} +$
$3\,919\,104\, y_m^{14} + 8\,398\,080\, x_m^2\, y_m^{14} + 1\,679\,616\, y_m^{16} -$
$672\, z_m^2 + 19\,728\, x_m^2\, z_m^2 - 245\,592\, x_m^4\, z_m^2 +$
$1\,821\,852\, x_m^6\, z_m^2 - 4\,412\,880\, x_m^8\, z_m^2 -$
$25\,649\,136\, x_m^{10}\, z_m^2 - 15\,396\,480\, x_m^{12}\, z_m^2 +$
$8\,398\,080\, x_m^{14}\, z_m^2 + 19\,728\, y_m^2\, z_m^2 + 169\,776\, x_m^2\, y_m^2\, z_m^2 -$
$3\,434\,076\, x_m^4\, y_m^2\, z_m^2 + 40\,680\,144\, x_m^6\, y_m^2\, z_m^2 -$
$33\,662\,304\, x_m^8\, y_m^2\, z_m^2 + 320\,806\,656\, x_m^{10}\, y_m^2\, z_m^2 -$
$62\,145\,792\, x_m^{12}\, y_m^2\, z_m^2 - 245\,592\, y_m^4\, z_m^2 -$
$3\,434\,076\, x_m^2\, y_m^4\, z_m^2 - 14\,370\,048\, x_m^4\, y_m^4\, z_m^2 -$
$97\,452\,720\, x_m^6\, y_m^4\, z_m^2 - 414\,865\,152\, x_m^8\, y_m^4\, z_m^2 -$
$156\,204\,288\, x_m^{10}\, y_m^4\, z_m^2 + 1\,821\,852\, y_m^6\, z_m^2 +$
$40\,680\,144\, x_m^2\, y_m^6\, z_m^2 - 97\,452\,720\, x_m^4\, y_m^6\, z_m^2 +$
$946\,743\,552\, x_m^6\, y_m^6\, z_m^2 - 99\,097\,344\, x_m^8\, y_m^6\, z_m^2 -$
$4\,412\,880\, y_m^8\, z_m^2 - 33\,662\,304\, x_m^2\, y_m^8\, z_m^2 -$
$414\,865\,152\, x_m^4\, y_m^8\, z_m^2 - 99\,097\,344\, x_m^6\, y_m^8\, z_m^2 -$
$25\,649\,136\, y_m^{10}\, z_m^2 + 320\,806\,656\, x_m^2\, y_m^{10}\, z_m^2 -$
$156\,204\,288\, x_m^4\, y_m^{10}\, z_m^2 - 15\,396\,480\, y_m^{12}\, z_m^2 -$
$62\,145\,792\, x_m^2\, y_m^{12}\, z_m^2 + 8\,398\,080\, y_m^{14}\, z_m^2 +$
$9432\, z_m^4 - 245\,592\, x_m^2\, z_m^4 + 1\,540\,134\, x_m^4\, z_m^4 -$
$9\,082\,368\, x_m^6\, z_m^4 + 119\,322\,720\, x_m^8\, z_m^4 -$
$41\,150\,592\, x_m^{10}\, z_m^4 + 16\,796\,160\, x_m^{12}\, z_m^4 -$
$245\,592\, y_m^2\, z_m^4 - 3\,434\,076\, x_m^2\, y_m^2\, z_m^4 -$
$14\,370\,048\, x_m^4\, y_m^2\, z_m^4 - 97\,452\,720\, x_m^6\, y_m^2\, z_m^4 -$
$414\,865\,152\, x_m^8\, y_m^2\, z_m^4 - 156\,204\,288\, x_m^{10}\, y_m^2\, z_m^4 +$
$1\,540\,134\, y_m^4\, z_m^4 - 14\,370\,048\, x_m^2\, y_m^4\, z_m^4 +$
$450\,766\,944\, x_m^4\, y_m^4\, z_m^4 - 290\,573\,568\, x_m^6\, y_m^4\, z_m^4 +$
$1\,264\,750\,848\, x_m^8\, y_m^4\, z_m^4 - 9\,082\,368\, y_m^6\, z_m^4 -$

$$
\begin{aligned}
&97\,452\,720\, x_m^2\, y_m^6\, z_m^4 - 290\,573\,568\, x_m^4\, y_m^6\, z_m^4 - \\
&797\,817\,600\, x_m^6\, y_m^6\, z_m^4 + 119\,322\,720\, y_m^8\, z_m^4 - \\
&414\,865\,152\, x_m^2\, y_m^8\, z_m^4 + 1\,264\,750\,848\, x_m^4\, y_m^8\, z_m^4 - \\
&41\,150\,592\, y_m^{10}\, z_m^4 - 156\,204\,288\, x_m^2\, y_m^{10}\, z_m^4 + \\
&16\,796\,160\, y_m^{12}\, z_m^4 - 37\,800\, z_m^6 + 1\,821\,852\, x_m^2\, z_m^6 - \\
&9\,082\,368\, x_m^4\, z_m^6 - 112\,091\,040\, x_m^6\, z_m^6 - \\
&19\,035\,648\, x_m^8\, z_m^6 + 18\,475\,776\, x_m^{10}\, z_m^6 + \\
&1\,821\,852\, y_m^2\, z_m^6 + 40\,680\,144\, x_m^2\, y_m^2\, z_m^6 - \\
&97\,452\,720\, x_m^4\, y_m^2\, z_m^6 + 946\,743\,552\, x_m^6\, y_m^2\, z_m^6 - \\
&99\,097\,344\, x_m^8\, y_m^2\, z_m^6 - 9\,082\,368\, y_m^4\, z_m^6 - \\
&97\,452\,720\, x_m^2\, y_m^4\, z_m^6 - 290\,573\,568\, x_m^4\, y_m^4\, z_m^6 - \\
&797\,817\,600\, x_m^6\, y_m^4\, z_m^6 - 112\,091\,040\, y_m^6\, z_m^6 + \\
&946\,743\,552\, x_m^2\, y_m^6\, z_m^6 - 797\,817\,600\, x_m^4\, y_m^6\, z_m^6 - \\
&19\,035\,648\, y_m^8\, z_m^6 - 99\,097\,344\, x_m^2\, y_m^8\, z_m^6 + \\
&18\,475\,776\, y_m^{10}\, z_m^6 - 155\,439\, z_m^8 - 4\,412\,880\, x_m^2\, z_m^8 + \\
&119\,322\,720\, x_m^4\, z_m^8 - 19\,035\,648\, x_m^6\, z_m^8 + \\
&16\,796\,160\, x_m^8\, z_m^8 - 4\,412\,880\, y_m^2\, z_m^8 - \\
&33\,662\,304\, x_m^2\, y_m^2\, z_m^8 - 414\,865\,152\, x_m^4\, y_m^2\, z_m^8 - \\
&99\,097\,344\, x_m^6\, y_m^2\, z_m^8 + 119\,322\,720\, y_m^4\, z_m^8 - \\
&414\,865\,152\, x_m^2\, y_m^4\, z_m^8 + 1\,264\,750\,848\, x_m^4\, y_m^4\, z_m^8 - \\
&19\,035\,648\, y_m^6\, z_m^8 - 99\,097\,344\, x_m^2\, y_m^6\, z_m^8 + \\
&16\,796\,160\, y_m^8\, z_m^8 + 680\,400\, z_m^{10} - 25\,649\,136\, x_m^2\, z_m^{10} - \\
&41\,150\,592\, x_m^4\, z_m^{10} + 18\,475\,776\, x_m^6\, z_m^{10} - \\
&25\,649\,136\, y_m^2\, z_m^{10} + 320\,806\,656\, x_m^2\, y_m^2\, z_m^{10} - \\
&156\,204\,288\, x_m^4\, y_m^2\, z_m^{10} - 41\,150\,592\, y_m^4\, z_m^{10} - \\
&156\,204\,288\, x_m^2\, y_m^4\, z_m^{10} + 18\,475\,776\, y_m^6\, z_m^{10} + \\
&3\,055\,968\, z_m^{12} - 15\,396\,480\, x_m^2\, z_m^{12} + \\
&16\,796\,160\, x_m^4\, z_m^{12} - 15\,396\,480\, y_m^2\, z_m^{12} -
\end{aligned}
$$

$$62\,145\,792\, x_m^2\, y_m^2\, z_m^{12} + 16\,796\,160\, y_m^4\, z_m^{12} +$$
$$3\,919\,104\, z_m^{14} + 8\,398\,080\, x_m^2\, z_m^{14} +$$
$$8\,398\,080\, y_m^2\, z_m^{14} + 1\,679\,616\, z_m^{16}\}$$

It is an algebraic surface total degree 16 and all exponents even, with coefficients as large as 10^{10}.

◄ | ►

Solving algebraically...

One question that I asked myself: is this too much surface? Does it factor nontrivially?

Easy to show it does not factor over the rationals.

Also has no absolute factorization (that is, a nontrivial factorization over any algebraic extension of the rationals). This determination used symbolic−numeric computation technology that are outside the realm of this talk.

Upshot is, we get no degree reduction on this surface. I show a way to obtain a contour plot. It required a bit of care to get something that looks reasonable, runs relatively fast, and does not take a huge amount of memory.

◄ | ►

Graphing the algebraic surface

So here, at last, is the Wetzel Pretzel.

```
g = Compile[{{xm, _Real}, {ym, _Real},
    {zm, _Real}}, Evaluate[implicit]];
g1[x_Real, y_Real, z_Real] := g[x, y, z]
cp1 = ContourPlot3D[g1[xm, ym, zm] == 0,
  {xm, 0, .6}, {ym, 0, .6}, {zm, 0, .6},
  MaxRecursion → 0, PlotPoints → 40,
  Mesh → False,
  ContourStyle →
   RGBColor[0.137255, 0.913725, 1]]
```

Out[1]=



◄ | ►

What have we found?

Recall that our Lagrange multiplier computation did not in any explicit way enforce our inequality constraints. We found a variety, one that happens to intersect real space. Is any of it correct once we add those constraints?

Geometric consideration #1: The far reaches of this surface are correct. They arise from vertex parameter values that are not near to the positive octant borders (that is, not near the coordinate axis "edges").

Conclusion #1: We found some of the bounding surface.

Geometric consideration #2: The surface above does not stop for inequality constraints.

Conclusion #2: We have more work to do.

◄ | ►

More of the boundary surface

We now realize that the edge constraints for the vertex parameters are what we missed. Specifically, when a vertex moves from the interior of an octant face onto a coordinate axis, it can no longer continue in that direction. We need to account for this in searching for the bounding surface.

Mathematically: This part of parameter space is in the discriminant variety (the part where inequality boundaries are hit).

Our equations are now straightforward. We simply impose conditions, one at a time, that each previously unconstrained vertex parameter be zero (this constraint means that the vertex will lie on a coordinate axis).

◄ | ►

More boundary...

```
allparams = Complement[Variables[polys],
   midpt];
boundary =
 Table[
  First[GroebnerBasis[
    Join[polys, {allparams[[j]]}],
    midpt, allparams,
    MonomialOrder → EliminationOrder]],
  {j, Length[allparams]}]
```

$$\{1 - 48\,x_m^2 + 864\,x_m^4 - 6912\,x_m^6 + 20\,736\,x_m^8 - 24\,y_m^2 + 864\,x_m^2\,y_m^2 - 10\,368\,x_m^4\,y_m^2 + 41\,472\,x_m^6\,y_m^2 + 144\,y_m^4 - 3456\,x_m^2\,y_m^4 + 20\,736\,x_m^4\,y_m^4 - 24\,z_m^2 + 648\,x_m^2\,z_m^2 - 5184\,x_m^4\,z_m^2 + 10\,368\,x_m^6\,z_m^2 - 144\,y_m^2\,z_m^2 + 864\,x_m^2\,y_m^2\,z_m^2 + 10\,368\,x_m^4\,y_m^2\,z_m^2 + 144\,z_m^4 - 864\,x_m^2\,z_m^4 + 1296\,x_m^4\,z_m^4,$$

$$1 - 48\,x_m^2 + 864\,x_m^4 - 6912\,x_m^6 + 20\,736\,x_m^8 - 24\,y_m^2 + 648\,x_m^2\,y_m^2 - 5184\,x_m^4\,y_m^2 + 10\,368\,x_m^6\,y_m^2 + 144\,y_m^4 - 864\,x_m^2\,y_m^4 + 1296\,x_m^4\,y_m^4 - 24\,z_m^2 + 864\,x_m^2\,z_m^2 - 10\,368\,x_m^4\,z_m^2 + 41\,472\,x_m^6\,z_m^2 - 144\,y_m^2\,z_m^2 + 864\,x_m^2\,y_m^2\,z_m^2 + 10\,368\,x_m^4\,y_m^2\,z_m^2 + 144\,z_m^4 - 3456\,x_m^2\,z_m^4 + 20\,736\,x_m^4\,z_m^4,$$

$$1 - 24\,x_m^2 + 144\,x_m^4 - 48\,y_m^2 + 864\,x_m^2\,y_m^2 - 3456\,x_m^4\,y_m^2 + 864\,y_m^4 - 10\,368\,x_m^2\,y_m^4 + 20\,736\,x_m^4\,y_m^4 - 6912\,y_m^6 + 41\,472\,x_m^2\,y_m^6 + 20\,736\,y_m^8 - 24\,z_m^2 - 144\,x_m^2\,z_m^2 + 648\,y_m^2\,z_m^2 + 864\,x_m^2\,y_m^2\,z_m^2 - 5184\,y_m^4\,z_m^2 + 10\,368\,x_m^2\,y_m^4\,z_m^2 + 10\,368\,y_m^6\,z_m^2 + 144\,z_m^4 - 864\,y_m^2\,z_m^4 + 1296\,y_m^4\,z_m^4,$$

$$1 - 24 x_m^2 + 144 x_m^4 - 48 y_m^2 + 648 x_m^2 y_m^2 - 864 x_m^4 y_m^2 + 864 y_m^4 - 5184 x_m^2 y_m^4 + 1296 x_m^4 y_m^4 - 6912 y_m^6 + 10\,368 x_m^2 y_m^6 + 20\,736 y_m^8 - 24 z_m^2 - 144 x_m^2 z_m^2 + 864 y_m^2 z_m^2 + 864 x_m^2 y_m^2 z_m^2 - 10\,368 y_m^4 z_m^2 + 10\,368 x_m^2 y_m^4 z_m^2 + 41\,472 y_m^6 z_m^2 + 144 z_m^4 - 3456 y_m^2 z_m^4 + 20\,736 y_m^4 z_m^4,$$

$$1 - 24 x_m^2 + 144 x_m^4 - 24 y_m^2 - 144 x_m^2 y_m^2 + 144 y_m^4 - 48 z_m^2 + 864 x_m^2 z_m^2 - 3456 x_m^4 z_m^2 + 648 y_m^2 z_m^2 + 864 x_m^2 y_m^2 z_m^2 - 864 y_m^4 z_m^2 + 864 z_m^4 - 10\,368 x_m^2 z_m^4 + 20\,736 x_m^4 z_m^4 - 5184 y_m^2 z_m^4 + 10\,368 x_m^2 y_m^2 z_m^4 + 1296 y_m^4 z_m^4 - 6912 z_m^6 + 41\,472 x_m^2 z_m^6 + 10\,368 y_m^2 z_m^6 + 20\,736 z_m^8,$$

$$1 - 24 x_m^2 + 144 x_m^4 - 24 y_m^2 - 144 x_m^2 y_m^2 + 144 y_m^4 - 48 z_m^2 + 648 x_m^2 z_m^2 - 864 x_m^4 z_m^2 + 864 y_m^2 z_m^2 + 864 x_m^2 y_m^2 z_m^2 - 3456 y_m^4 z_m^2 + 864 z_m^4 - 5184 x_m^2 z_m^4 + 1296 x_m^4 z_m^4 - 10\,368 y_m^2 z_m^4 + 10\,368 x_m^2 y_m^2 z_m^4 + 20\,736 y_m^4 z_m^4 - 6912 z_m^6 + 10\,368 x_m^2 z_m^6 + 41\,472 y_m^2 z_m^6 + 20\,736 z_m^8\}$$

◄ | ►

More boundary...

```
cp2 = ContourPlot3D[Evaluate[boundary == 0],
  {xm, 0, .6}, {ym, 0, .6}, {zm, 0, .6},
  MaxRecursion → 0, PlotPoints → 30,
  Mesh → None, ContourStyle → Lighter[Red],
  Lighting → "Neutral"]
```

Out[2]=



◄ | ►

The solid between the surfaces

We want to visualize this bounding surface. First we show slices in two dimensions. For this we will use both the algebraic surfaces we computed, and the crudely approximated geometric region. This fit is imperfect, largely due to the approximate nature of that region plot.

```
topslices = Table[boundary,
   {xm, .015, .6, .07}];
topplots =
  Map[ContourPlot[# == 0, {ym, 0, .6},
     {zm, 0, .6}, MaxRecursion → 1,
     PlotPoints → 25,
     ColorFunction →
      Function[{x, y, f}, Red],
     ContourStyle → {Thickness[.004]}] &,
   topslices];
```

```
bottomslices =
  Table[implicit, {xm, .015, .6, .07}];
bottomplots =
  Map[ContourPlot[# == 0, {ym, 0, .6},
     {zm, 0, .6}, MaxRecursion → 1,
     PlotPoints → 25,
     ColorFunction →
      Function[{x, y, f}, Blue],
     ContourStyle →
      {Thickness[.004], Dashed}] &,
   bottomslices];
```

```
midslices = Table[xm, {xm, .015, .6, .07}];
midplots =
  Map[RegionPlot[inRegion[{#, y, z}, .015],
     {y, 0, .6}, {z, 0, .6},
     ColorFunction →
      Function[{x, y, z}, Green],
     MaxRecursion → 0, PlotPoints → 10] &,
   midslices];
```

◄ | ►

The solid...

```
Map[Show,
 Transpose[{midplots, topplots,
   bottomplots}]]
```




0.6
0.5
0.4
0.3
0.2
0.1
0.0
0.0 0.1 0.2 0.3 0.4 0.5 0.6

The solid in 3D

In which we use opacity, squint our eyes, and more or less visualize our region.

```
Graphics3D[{{Opacity[0.35], First[cp1]},
  {Opacity[0.55], First[cp2]}}]
```





◄ | ►

So there we are

We have, in some sense, found our solid.

What more could we want?

- Faster, better graphics. The contour plots in 3D are expensive to compute accurately, and slow to zoom/rotate if done to high precision.
- A way to remove self intersecting parts of the surfaces
- Better ways to visualize the solid between the surfaces. (I suspect this can be done just fine with available graphics technology. But not by me.)

◄ | ►

Summary

We developed some tools for computing and visualizing the triangle−in−a−corner midpoint locus. These use a melange of methods from symbolic, numeric, computational geometry, and graphical computation. In my day job I am one of the toolsmiths. Though not for graphics, on which I claim illiteracy.

One can push this work in various directions. I'll mention a few.

- Tackle related math problems. It might be useful to understand which ones are amenable to the methods we showed. This in turn might give insight into...
- ...Generalizing to problems of technological interest. Often these show even worse behavior than the idealized math problems.
- Work on the "inverse problem": Given a starting configuration and a new allowed midpoint value, find a path in the parameter space that gets us to the new configuration. This amounts to setting up and solving a system of differential algebraic equations. Probably simple when path connecting initial and final midpoint location is a line segment. Might get more intricate when it is some other parametrized curve (the sort of issue encountered in motion planning in presence of obstacles, or when direction changes need to be smooth).